

Kombinierte Strahlung und Wärmeleitung durch ein Medium

Ausgabe: 8 / 2002

Problem:

Innerhalb eines geschlossenen Gehäuses soll der Wärmetransport durch Wärmeleitung und Strahlung berechnet werden.




Erläuterung:

Will man die Temperaturverteilung in einem geschlossenen mit Luft gefüllten Behälter untersuchen, so müssen neben der meist kleinen Wärmeleitung in der Luft auch Konvektion und Strahlung berücksichtigt werden. Während die Berechnung der Konvektion eine Strömungsberechnung mit ANSYS/FLOTRAN erforderlich macht, lassen sich Wärmeleitung und Strahlung mit den normalen ANSYS-Temperaturfeldelementen berechnen.

Es gibt im ANSYS zwei Methoden zur Berechnung der Strahlung, die Radiosity Methode und die Matrixmethode. Nur letztere kann zur Modellierung der Strahlung innerhalb eines Berechnungsgebietes verwendet werden. Die Strahlung austauschenden Ränder werden durch link32 bzw. shell57 Elemente identifiziert. Mit Hilfe dieser Elemente wird das Superelement aufgestellt, welches dann den Strahlungsaustausch zwischen den betroffenen Knoten liefert. Dabei ist es völlig unerheblich zwischen welchen Knoten der Strahlungsaustausch stattfindet, nach Aussen oder nach Innen.

Beispiel:

Betrachtet wird ein einfacher viereckiger Raum, der von einer Wand umschlossen wird. Im Inneren ist die Wärmeleitung sehr viel kleiner als in der umgebenden Wand. Auf der linken Seite wird die Temperatur auf 100 Grad gehalten, während auf der rechten Seite ein Waermestrom von 1000 W/m² aufgeprägt wird. In den resultierenden Temperaturverteilungen erkennt man sehr deutlich den grossen Einfluss der Strahlung. Ohne Strahlung ist die Temperatur sehr viel höher, weil der Wärmetransport fast ausschliesslich infolge von Wärmeleitung durch die Wand erfolgt.



Kombinierte Strahlung und Wärmeleitung durch ein Medium

Ausgabe: 8 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
!*****  
!Strahlung und Wärmeleitung  
!im Inneren eines Bauteils  
!Strahlung mit Matrixmethode  
!*****
```

```
fini  
/clear  
/filna,radiation
```

```
/prep7
```

```
rect,0,1,0,1  
rect,-0.1,1.1,-0.1,1.1  
aovl,all
```

```
asel,s,,1  
aatt,1  
asel,s,,3  
aatt,2
```

```
alls
```

```
et,1,55
```

```
esiz,0.05  
amesh,all
```

```
mp,kxx,1,0.01  
mp,kxx,2,20
```

```
et,2,32  
type,2  
lsel,s,,1,4,1  
nsl,s,1  
lmesh,all
```

```
nsl,s,loc,x,-0.1  
d,all,temp,100
```

```
nsl,s,loc,x,1.1  
!d,all,temp,400  
sf,all,hflux,1000
```

```
alls
```

```
!Strahlung  
/aux12  
lsel,s,,1,4,1  
nsl,s,1  
esln,s,1  
emis,1,0.8  
emis,2,0.8  
stef,5.67e-8  
geom,1  
vtype,1,200  
write,strahl,sub  
alls
```

```
/prep7  
et,3,50  
keyopt,3,1,1  
type,3  
se,strahl,sub
```

```
tunif,200  
toffs,273
```

```
esel,u,ename,,32
```

```
/solu  
solve
```

```
/post1  
set,last  
plnsol,temp
```